
Trajectoire de Hill

La trajectoire de Hill est liée au problème des trois corps restreint. Il s’agit d’une trajectoire périodique d’un
corps de masse nulle dans le champ de gravitation de deux corps massifs. Une illustration de ce problème
peut-être vue dans la trajectoire d’un satellite artificiel dans le champ de gravitation du couple Terre-Lune.
Définition des masses relatives des corps massifs de sorte que la masse totale soit égale à 1. Les valeurs
utilisées ici correspondent, à peu près, au couple Terre-Lune.
--> m=1/82.45
m =

0.0121286
--> M=1-m
M =

0.9878714
De ce fait, la Terre est en (−m, 0) et la Lune en (M , 0).
Le problème est plan et, dynamiquement, l’état du satellite (u dans la suite) est caractérisé par 4 variables,
deux pour la position et deux pour le moment.
Calcul du cube de la distance du satellite à la Terre.
--> function [r] = d1(u)
--> r = ((u(1)+m)^2+u(2)^2)^(3/2)
--> endfunction

Calcul du cube de la distance du satellite à la Lune.
--> function [r] = d2(u)

--> r = ((u(1)- M)^2+u(2)^2)^(3/2)
--> endfunction

Maintenant voici le second membre des équations différentielles régissant le mouvement du satellite.
--> function [f] = Ham3Corps(t,u)
--> f(1) = u(3) + u(2)
--> f(2) = u(4) - u(1)
--> f(3) = u(4) - M *(u(1)+m)/d1(u) - m *(u(1)-M)/d2(u)
--> f(4) = -u(3) - u(2) * (M/d1(u)+m/d2(u))
--> endfunction

Les conditions initiales propres à obtenir la trajectoire de Hill :
--> u0 = [1.2;0;0;1.2 - 1.04935750483]
u0 =

1.2
0
0

0.1506425
Une subdivision de la période du mouvement :
--> t = linspace(0,6.19217,10000);

Intégration du système, calcul des positions du satellite aux points de la subdivision :
--> [U] = ode(u0,0,t,Ham3Corps);

Tracé de la trajectoire :
--> plot2d(U(1,:)’ ,U(2,:)’,2,"111","",[-1.5,-1.06,1.5,1.06])

Placement de la Terre et de la Lune :
--> plot2d(-m,0,-3)
--> plot2d(M,0,-9)

